

CMake Cheatsheet – A gentle introduction to CMake

This cheatsheet will give you an idea how CMake works and how it can be used to configure software projects.

CMake - Creating a simple C++ project

CMake is a tool that allows users to configure how a source code project should be built on a specific platform.

A small project could be organized like this (Example1):

```
CMakeLists.txt
src/main.cpp
src/foo.cpp
src/foo.hpp
```

This project contains two source files located in the src directory and one header file in the include directory in the same directory.

When running CMake on this project you are asked to for a binary directory. It is best practise to create a new directory, since this directory will contain all files related to building the project. If something goes wrong, you can delete the folder and start over.

Running CMake will not create the final executable, but instead it will generate project files for Visual Studio, XCode or makefiles. Use these tools to build the project.

Understanding CMakeLists.txt

Creating project files using CMake requires a `CMakeLists.txt` file, which describes how the project is structured and how it should be build.

For example 1 the file looks like this:

```
cmake_minimum_required (VERSION 2.9)

# Setup projectname
project (HelloProject)

# Compile and link main.cpp and foo.cpp
# into the executable Hello
add_executable(Hello src/main.cpp src/foo.cpp)
```

First the minimum version of CMake is defined. Then the project name is defined using the command `project()`. A project can contain multiple targets (either executables or libraries). This project defines a single executable target called `Hello`, which is created by compiling and linking the two source files `main.cpp` and `foo.cpp` files.

When the two source files are compiled the compiler will search for the header file `foo.h` since both source files depend on this using a `#include "foo.hpp"`. Since the file is located in the same located as the source file, the compiler will not have any problems finding the file.

The CMake Scripting Language

The `CMakeLists.txt` describes the build process using a command based programming language. The commands are case insensitive and take a list of arguments.

```
# This is a comment.
COMMAND( arguments go here )
ANOTHER_COMMAND() # this command has no arguments
YET_ANOTHER_COMMAND( these
                      arguments are spread      # another comment
                      over several lines )
```

CMake script also has variables. Variables can either be defined by CMake or can be defined in the CMake script. The command `set(parameter value)` set a given parameter to a value. The command `message(value)` print out the value to the console. To get the value of a variable use `${varname}`, which substitutes the variable name with its value.

```
cmake_minimum_required (VERSION 2.9)

SET( x 3 )      # x = "3"
SET( y 1 )      # y = "1"
MESSAGE( x y )  # displays "xy"
MESSAGE( ${x}${y} ) # displays "31"
```

All variable values are a text string. Text strings can be evaluated as boolean expressions (e.g. when used in `IF()` and `WHILE()`). The values `"FALSE"`,

"OFF", "NO", or any string ending in "-NOTFOUND" evaluates be false - everything else to true.

Text strings can represent multiple values as a list by separating entities using semi-colon.

```
cmake_minimum_required (VERSION 2.9)

SET( x 3 2) # x = "3;2"
SET( y hello world !) # y = "hello;world;!"
SET( z "hello_world!" ) # y = "hello world!"
MESSAGE( ${x} ) # prints "xy"
MESSAGE( "y=_${y}_z=_${z}" )
# prints y = hello;world;! z = hello world!
```

Lists can be iterated using the command **FOREACH** (var val):

```
cmake_minimum_required (VERSION 2.9)

SET( x 3 2) # x = "3;2"
FOREACH ( val ${x})
    MESSAGE( ${val} )
ENDFOREACH( val)

# prints:
# 3
# 2
```

Exposing compile options

CMake allows the end user (who runs CMake) to modify some values of some variables. This is usually used to defined properties of the build such as locations of files, machine architecture and string values.

The command **set(<variable> <value> CACHE <type> <docstring>)** set the variable to the default value - but allows the value to be changed by the cmake user when configuring the build. The type should be one of the following:

- FILEPATH = File chooser dialog.
- PATH = Directory chooser dialog.
- STRING = Arbitrary string.


- BOOL = Boolean ON/OFF checkbox.
- INTERNAL = No GUI entry (used for persistent variables).

In the following example the user can configure if "Hello" or an alternative string should be printed based on the configuration variables hello and other_msg.

```
cmake_minimum_required (VERSION 2.9)

SET(hello true CACHE BOOL "If_true_write_hello")
SET(other_msg "Hi" CACHE STRING "Not_hello_value")
IF ( ${hello} )
    MESSAGE( "Hello" )
ELSE ( ${hello} )
    MESSAGE( ${other_msg} )
ENDIF ( ${hello} )
```

During configuration of the project, the CMake user get prompted with the exposed options.



The values that the CMake user enters will be saved in the text file **CMakeCache.txt** as key-value pairs:

```
// ....
//Print hello
hello:BOOL=OFF

//Not hello value
other_msg:STRING=Guten tag
// ....
```

Complex projects

Some project both contains multiple executables and multiple libraries. This is project structure for instance when having both unit tests and programs. It is common to separate these sub projects into subfolders. Example:

```
CMakeLists.txt
somelib/CMakeLists.txt
somelib/foo.hpp
somelib/foo.cpp
someexe/CMakeLists.txt
someexe/main.cpp
```

The main **CMakeLists.txt** contains the basic project settings but then includes the sub projects:

```
# CMakeLists.txt
cmake_minimum_required (VERSION 2.9)

# Setup projectname
project (HelloProject)

add_subdirectory(somelib)
add_subdirectory(someexe)
```

First the library Foo is compiled from the source in the **somelib** directory:

```
# somelib/CMakeLists.txt

# Compile and link foo.cpp
add_library(Foo foo.cpp)
```

Finally, the executable Hello is compiled and linked to the Foo library. Since the main.cpp references to header file Foo.hpp the somelib directory is added to the header search path:

```
# someexe/CMakeLists.txt

# add somelib to header search path
include_directories(.. /somelib/)

add_executable(Hello main.cpp)

# link to Foo library
target_link_libraries(Hello Foo)
```

Runtime resources

Often runtime resources (such as DLLs, game-assets and text files) are read relative to the executable. One solution is to copy resources into the same directory as the executable. Example:

```
CMakeLists.txt
someexe/main.cpp
someexe/res.txt
```

In this project the source files assumes that the resource is located in the same directory as the executable:

```
// main.cpp
#include <iostream>
#include <fstream>

int main(){
    std::fstream f("res.txt");
    std::cout << f.rdbuf();
    return 0;
}
```

The CMakeLists.txt make sure to copy the file.

```
# CMakeLists.txt
cmake_minimum_required (VERSION 2.9)
```

```
# Setup projectname
project (HelloProject)

add_executable(Hello someexe/main.cpp)

file(COPY someexe/res.txt DESTINATION Debug)
file(COPY someexe/res.txt DESTINATION Release)
```

Note: One problem with this approach is if you modify the original resources, then you need to run CMake again.

Links and information

https://cmake.org/Wiki/CMake/Language_Syntax

<https://cmake.org/cmake/help/v3.0/command/set.html>

Created by Morten Nobel-Jørgensen, mnob@itu.dk, ITU, 2017

Released under the MIT license.

Latex template by John Smith, 2015

<http://johnsmith.com/>

Released under the MIT license.